

Bouw je eigen weerstation: deel 1

Het meten van de temperatuur en de luchtvochtigheid.

Om de temperatuur en luchtvochtigheid te meten gebruiken we een opnemer (ook wel sensor genoemd) DHT22. Op de volgende plaatjes een afbeelding de DHT22.



Een losse DHT22



De DHT22 gemonteerd op een printje


Het aansluiten van de DHT22 op de Arduino:

Vcc → 5 volt van de Arduino


GND → ground van de Arduino

Data → pen 9 van de Arduino

Note: Bij een losse DHT22 moet er een weerstand van 4700 Ohm tussen Vcc en data worden opgenomen; bij een printje is dat niet nodig omdat de weerstand al op het printje is gemonteerd.



Het schema



De proefopstelling

Het programma

Het Arduino programma zal de temperatuur en luchtvochtigheid meten en daarna deze waarden via de seriële monitor op het scherm van de computer laten zien.

Om met de sensor DHT22 te kunnen werken moet eerst de bibliotheek van deze sensor aan de Arduino bibliotheek worden toegevoegd. Toevoegen gaat als volgt: -> IDE menu -> Schets -> Bibliotheek gebruiken -> Bibliotheek beheren -> DHT22 in zoekvenster schrijven en

nu laten zoeken -> DHT sensor library by ADAfruit kiezen -> installeren (indien melding: Adafruit Unified Sensor -> Install all)

Nu kunnen we in ons Arduino programma de DHT22 sensor gebruiken.

In het programma maken we ook kennis met variabelen van het type 'float'. Dit zijn getallen met een komma; bijvoorbeeld 23,4. Dit is nodig omdat we de temperatuur en de luchtvochtigheid heel precies willen meten.

De code

```
// sketch weerstation 1a.
// meten van de relatieve luchtvochtigheid en temperatuur met een DHT22 sensor.
// de meetwaarden wordt naar de seriële monitor gestuurd.

#include <DHT.h>; // DHT sensor bibliotheek toevoegen
// DHT22 type van de sensor en pen 9 voor data

DHT sensor_dht(9, DHT22); // onze sensor heet nu: sensor_dht


float temperatuur, vochtigheid; // 2 variabelen om de meetwaarden op te slaan

void setup()
{
  sensor_dht.begin(); // start de sensor om metingen te doen
  Serial.begin(9600); // start de seriële monitor
}

void loop()
{
  delay(2000); // wacht 2 seconden
  //lees de sensor (temperatuur en de vochtigheid) uit en bewaar die in de variabelen
  temperatuur = sensor_dht.readTemperature();
  vochtigheid = sensor_dht.readHumidity();

  // stuur de gemeten waarde naar de seriële monitor
  Serial.print("Luchtvochtigheid: ");
  Serial.print(vochtigheid);
  Serial.print(" %, Temperatuur: ");
  Serial.print(temperatuur);
  Serial.println(" Celsius");
}
```

De seriële monitor



```
COM5
Lucht vochtigheid: 56.50 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.30 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.10 %, Temperatuur: 30.70 Celsius
Lucht vochtigheid: 57.40 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.60 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.90 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.90 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.80 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.60 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.50 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 57.30 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 62.10 %, Temperatuur: 30.80 Celsius
Lucht vochtigheid: 98.40 %, Temperatuur: 31.10 Celsius
Lucht vochtigheid: 97.70 %, Temperatuur: 31.00 Celsius
Lucht vochtigheid: 87.70 %, Temperatuur: 30.90 Celsius
Lucht vochtigheid: 76.40 %, Temperatuur: 30.90 Celsius
Autoscroll Show timestamp
```

Uitbreiding met LCD

Probeer nu zelf om de temperatuur en de luchtvochtigheid op een LCD schermpje te schrijven. Hint: kijk eens naar de code van oefening 4 (het LCD) van werkbladen start.

<https://on4cdu.net/wp-content/uploads/Coderdojo/Tervuren/Start/werkbladen-start-Coderdojo-Tervuren-v0.1.pdf>

Kom je er niet uit dan kijk eens naar de code hieronder:

```
// sketch weerstation 1b.
// Meten van vochtigheid en temperatuur met een DHT22 sensor.
// de meetwaarde wordt naar de seriële monitor gestuurd en op het LCD getoond.

#include <DHT.h>; // DHT sensor bibliotheek toevoegen
#include <LiquidCrystal.h>
// DHT22 type van de sensor
// pin 9 voor data

DHT sensor_dht(9, DHT22); // onze sensor heet nu: sensor_dht

// LCD Enable pin aan digital pin 11
// LCD R/W pin aan ground
// LCD D4 pin aan digital pin 5
// LCD D5 pin aan digital pin 4
// LCD D6 pin aan digital pin 3
// LCD D7 pin aan digital pin 2

LiquidCrystal lcd(12, 11, 5, 4, 3, 2);

float temperatuur, vochtigheid; // 2 variabelen om de meetwaarden op te slaan

void setup()
{
  sensor_dht.begin(); // start de sensor om metingen te doen
  Serial.begin(9600); // start de seriële monitor
  lcd.begin(16,2); // start LCD type 16 x 2
}

void loop()
{
  delay(2000); // wacht 2 seconden
  //lees de temperatuur en de vochtigheid uit en bewaar die in de variabelen
  temperatuur = sensor_dht.readTemperature();
  vochtigheid = sensor_dht.readHumidity();

  // stuur de gemeten waarden naar de seriële monitor
  Serial.print("Luchtvochtigheid: ");
  Serial.print(vochtigheid);
  Serial.print(" %, Temperatuur: ");
  Serial.print(temperatuur);
  Serial.println(" Celsius");

  // nu de metingen op het LCD vertonen
  lcd.setCursor(0, 0); // zet cursor op positie 0 (vooraan) en rij 0 (eerste regel)
  lcd.print ("Temp : "); lcd.print(temperatuur,1); lcd.print (" C ");
```

```
lcd.setCursor(0,1); // zet cursor op positie 0 (vooraan) en rij 1 (tweede regel)
lcd.print ("Vocht: "); lcd.print(vochtigheid,1); lcd.print (" % ");
}
```

